

Calculating half-life for salicylic acid clearance

The ordinary differential equation for clearance is given by

$$dSA/dt = -k \cdot SA$$

with rate constant k and with initial condition

$$SA(t=0) = SA_0.$$

The solution is given as

$$SA(t) = SA_0 \cdot \exp(-k \cdot t).$$

When half of the salicylic acid is removed, the equation becomes

$$SA_0/2 = SA_0 \cdot \exp(-k \cdot t_{half}).$$

Dividing both sides by SA_0 and taking the natural logarithm of both sides yields

$$k = \log_e(2)/t_{half}.$$

Therefore, we can write the ordinary equation as

$$dSA/dt = -\log_e(2)/t_{half} \cdot SA.$$

The model in JSim is as follows:

```
import nsrunit;
unit conversion on;

math halfLife {
  realDomain t hour ; t.min=0; t.max=16; t.delta=0.1;
  real SA(t) mg/L, SA0 =1 mg/L;
  real thalf = 2 hour;
  when(t=t.min) SA=SA0;
  SA:t = -ln(2)/thalf*SA;
}
```

An alternate version is

```
import nsrunit;
unit conversion on;

math halfLife {
  realDomain t hour ; t.min=0; t.max=16; t.delta=0.1;
  real SA(t) mg/L, SA0 =1 mg/L;
  real k = 0.3465735903 hour(-1);
  real thalf = ln(2)/k;
  when(t=t.min) SA=SA0;
  SA:t = -k*SA;
}
```

Does it make a difference which model we use? The first model can give us error bars on k, the second model can give us error bars on half life.