
Pharmacokinetics of salicylic acid: Halflife in Humans
1 mg SA/L = .724E5 M = 7.24 uM. 1 mole SA = 138.1 grams.
Low Dose Data:

(IH Benedek, AS Joshi, JH Pieniazek, SY King and DM Kornhauser; Variability in the 
pharmacokinetics and pharmacodynamics of low dose aspirin in healthy male volunteers.
J. Clin. Pharmacl 1995; 35; 1181)
The data for the low dose model was taken from IH Benedek, et al.,
    Figure 1, right panel, last seven points dose period 1.

Midrange Dose Data:

L Aarons, K Hopkins, M Rowland, S. Brossel, and JF Thiercelin: Route of administration 
and sex differences in the pharmacokinetics of aspirin, administered as its lysine salt.

Pharmaceutical Research, Vol 6, No 8, 1989. p 660666.
The data for the mid dose model was  taken from L Aarons, et al.,
    the last eleven points of the orally administered does.

High Dose Data:

(LF Prescott, M BalaliMood, JAJH Critchley, AF Johnstone, AT Proudfood; Diuresis or 
urinary alkalinisation for salicylate poisoning? British Medical Journal,  285 13 
November 1982 p1383 1384.) The data from the high does model was taken 
from LF Prescott, et al., the last nine points from the Control.

Digitized Data:

The same 
data 
plotted semilogarithmically:

Calculating halflife for salicylic acid clearance

The ordinary differential equation is for clearance  is given by

  dSA / dt=−k⋅SA

with rate constant k and with initial condition

  SAt=0=SA0.
The solution is given as

  SAt =SA0⋅exp−k⋅t .

When half of the salicylic acid is removed, the equation becomes

  SA0/2=SA0⋅exp−k⋅t half .

Dividing both sides by SA0 and taking  the natural logarithm of both sides 
yields

  k=loge2/ t half .

Therefore, we can write the ordinary equation as

  dSA/dt=−loge 2/ t half⋅SA.

// The model in JSim is as follows:

import nsrunit; 
unit conversion on;

math halfLife {
realDomain t hour ; t.min=0; t.max=16; t.delta=0.1;
real SA(t) mg/L, SA0 =1 mg/L;
real thalf = 2 hour;
when(t=t.min) SA=SA0;
SA:t = -ln(2)/thalf*SA;
}

// An alternate version is

import nsrunit; 
unit conversion on;

math halfLife {
realDomain t hour ; t.min=0; t.max=16; t.delta=0.1;
real SA(t) mg/L, SA0 =1 mg/L;
real k = 0.3465735903 hour^(-1);
real thalf = ln(2)/k;
when(t=t.min) SA=SA0;
SA:t = -k*SA;
}

Does it make a difference which model we use? The first model can give us 
error bars on k, the rate constant, the second model can give us error bars on 
half life.

//Here is a simple enzyme conversion model:

/* Simple Enzyme model for data
 
            kf1->                  kf2->
   S+Enzyme<----->  SEnzymeComplex<-----> Enzyme+Product
            <-kb1                  <-kb2
   
*/
import nsrunit; unit conversion on;
math SimpleEnzyme { 

realDomain t hour; t.min=0; t.max=16.0; t.delta=0.05;
real kf1  = 10. L/(mg*hour), kb1  = 10. hour^(-1); // rate constants 
for substrate
real kb2  = 0.0 L/(mg*hour), kf2  = 10. hour^(-1); // rate constants 
for product 
real Etot = 0.1 mg/L;                              // Total Amount of 
Enzyme

real S(t) mg/L, E(t) mg/L, SE(t) mg/L, P(t) mg/L; 
real Stot = 1 mg/L;

when(t=t.min) {S=Stot; SE=0;}
S:t  = -kf1*S*E + kb1*SE;
SE:t =  kf1*S*E - kb1*SE - kf2*SE + kb2*E*P;
Stot = S + SE + P ;
Etot = E + SE ;

}

/* Enzyme model for all sets of data
   using shared parameters
 
            kf1->                  kf2->
   S+Enzyme<----->  SEnzymeComplex<-----> Enzyme+Product
            <-kb1                  <-kb2
   
*/

import nsrunit;
unit conversion on;

math FullEnzyme {
realDomain t hour; t.min=0; t.max=16.0; t.delta=0.05;
real kf1  = 10. L/(mg*hour), kb1  = 10. hour^(-1); // rate constants 
                                                   // for substrate
real kb2  = 0.0 L/(mg*hour), kf2  = 10. hour^(-1); // rate constants 
                                                   // for product
real Etot = 0.1 mg/L;                              // Total Amount
                                                   // of Enzyme

real LS(t) mg/L, LE(t) mg/L, LSE(t) mg/L, LP(t) mg/L;
real LStot = 1 mg/L;
when(t=t.min) {LS=LStot; LSE=0;}
LS:t  = -kf1*LS*LE + kb1*LSE;
LSE:t =  kf1*LS*LE - kb1*LSE - kf2*LSE + kb2*LE*LP;
LStot = LS + LSE + LP ;
Etot = LE + LSE ;

real MS(t) mg/L, ME(t) mg/L, MSE(t) mg/L, MP(t) mg/L;
real MStot = 1 mg/L;
when(t=t.min) {MS=MStot; MSE=0;}
MS:t  = -kf1*MS*ME + kb1*MSE;
MSE:t =  kf1*MS*ME - kb1*MSE - kf2*MSE + kb2*ME*MP;
MStot = MS + MSE + MP ;
Etot = ME + MSE ;

real HS(t) mg/L, HE(t) mg/L, HSE(t) mg/L, HP(t) mg/L;
real HStot = 1 mg/L;
when(t=t.min) {HS=HStot; HSE=0;}
HS:t  = -kf1*HS*HE + kb1*HSE;
HSE:t =  kf1*HS*HE - kb1*HSE - kf2*HSE + kb2*HE*HP;
HStot = HS + HSE + HP ;
Etot = HE + HSE ;
}

//MODEL with Briggs-Haldane Kinetics

import nsrunit; unit conversion on;
math SalicylicAcidClearance { realDomain t hour; t.min=0; t.max=16.0; 
t.delta=0.05;
/*
 A Mathematical Model expressing clearance as exponential decay

 C(t)=C0*exp(-k*t)

Half Life:
 C0/2=C0*exp(-k*thalf),    Divide both sides by C0 and take the natural 
log
 ln(1/2)=ln(1)-ln(2)=-ln(2)=-k*t. Therefore
 k=ln(2)/thalf;

Instead of writing dC/dt=-k*C, we will write

 dC/dt=-(ln(2)/thalf)*C.
 
 

*/
real lo(t) mg/L, mid(t) mg/L, hi(t) mg/L;
 lo.desc  ="lo concentration";

mid.desc ="mid concentration";
hi.desc  ="hi concentration";

real Alo = 1 mg/L, Amid =50 mg/L, Ahi =300 mg/L;
 Alo.desc  ="Initial lo concentration";

Amid.desc ="Initial mid concentration";
Ahi.desc  ="Initial hi concentration";

real thalflo =5 hour, thalfmid = 5 hour, thalfhi = 5 hour;
 thalflo.desc="half life lo";
 thalfmid.desc="half life mid";

thalfhi.desc="half life hi";
when(t=t.min) {lo=Alo; mid=Amid; hi=Ahi; }
lo:t=(-ln(2)/thalflo)*lo;
mid:t=(-ln(2)/thalfmid)*mid;
hi:t=(-ln(2)/thalfhi)*hi;

/* Enzyme model for all sets of data
   using shared parameters
 
                  kf1                      kf2
  SA+Enzyme --> SAEnzymeComplex --> Enzyme+Product
                  kb1
   
*/
real kf1= 1 L*mg^(-1)*hour^(-1), kb1 = 1 hour^(-1), kf2  = 1 hour^(-1);

kf1.desc="first forward rate constant";

kf2.desc="second forward rate constant";
real LOenz(t)  mg/L,  LOE(t)  mg/L,  Elo(t)  mg/L;
real MIDenz(t) mg/L,  MIDE(t) mg/L,  Emid(t) mg/L;
real HIenz(t)  mg/L,  HIE(t)  mg/L,  Ehi(t)  mg/L;
 
real Etot = 6 mg/L;
when(t=t.min) {LOenz = Alo;  MIDenz = Amid; HIenz = Ahi;
               LOE   = 0;    MIDE   = 0;    HIE   = 0;}

LOenz:t  =   -kf1*LOenz*Elo+kb1*LOE;
LOE:t    = -(-kf1*LOenz*Elo+kb1*LOE+kf2*LOE);
Elo      = Etot-LOE; 

MIDenz:t =   -kf1*MIDenz*Emid+kb1*MIDE;
MIDE:t   = -(-kf1*MIDenz*Emid+kb1*MIDE+kf2*MIDE);
Emid     = Etot-MIDE; 

HIenz:t  =   -kf1*HIenz*Ehi+kb1*HIE;
HIE:t    = -(-kf1*HIenz*Ehi+kb1*HIE+kf2*HIE);
Ehi      = Etot-HIE; 

/*-----------Briggs-Haldane Enzyme Kinetics-------------

   dC/dt = -Vmax*C/(Km+C)

*/
real estKm mg/L;
estKm = (kf2+kb1)/kf1;
real estVmax mg/L*hour^(-1);
estVmax = kf2*Etot;

real LOBH(t) mg/L, MIDBH(t) mg/L, HIBH(t) mg/L;
when(t=t.min) {LOBH=Alo; MIDBH=Amid; HIBH=Ahi;}
real Km = 10 mg/L;
real Vmax = 8 mg/(L*hour);

LOBH:t=-Vmax*LOBH/(Km+LOBH);
MIDBH:t=-Vmax*MIDBH/(Km+MIDBH);
HIBH:t=-Vmax*HIBH/(Km+HIBH);

}

// Installed Model On Web Site:

/* MODEL NUMBER 280
    MODEL NAME: Aspirin
    SHORT DESCRIPTION: Salicylic acid (SA) clearance for three different
    dose ranges is modeled as an enzyme reaction.
 */

 import nsrunit; unit conversion on;

 math Aspirin {

 // INDEPENDENT VARIABLE
 realDomain t hour; t.min=0; t.max=90.0; t.delta=0.1;

 // PARAMETERS (SHARED BETWEEN 3 MODELS

 real kon1  = .1827 L*mg^(-1)*hour^(-1), // On rate for SA + enzyme
      KD1   = 6.304 mg/L,                // Dissociation constant for 
SA enzyme complex
      koff1 = KD1*kon1,                  // Off rate for SA enzyme 
complex
      kon2  = .0030 L/(mg*hour),         // On rate for Product + 
enzyme 
      KD2   = 250 mg/L,                  // Dissociation constant for 
Product enzyme complex
      koff2 = KD2*kon2,                  // Off rate for Product enzyme 
comples
      Gp    = 0.03 1/hour,               // Clearance rate from plasma
      Etot  = 10 mg/L;                   // Total enzyme

 /* LOW DOSE MODEL
    The data for the low dose model was taken from IH Benedek, et al.,
    Figure 1, right panel, last seven points dose period 1.
*/
 //     LOW DOSE PARAMETER
 real LSAtot = 8.183 mg/L,               // Total Low Dose concentration
 //     LOW DOSE MODEL VARIABLES
      LSA(t)  mg/L,                      // Low dose SA
      LSAE(t)  mg/L,                     // Low dose SA enzyme complex
      LE(t)  mg/L,                       // Low dose free enzyme
      LP(t) mg/L;                        // Low dose product
 //     LOW DOSE INITIAL CONDITIONS
 when(t=t.min) {LSA=LSAtot; LSAE=0; LP=0;}
 //     LOW DOSE ORDINARY DIFFERENTIAL AND MASS BALANCE EQUATIONS
 LSA:t = -kon1*LSA*LE+koff1*LSAE;
 LSAE:t = kon1*LSA*LE-koff1*LSAE-koff2*LSAE+kon2*LE*LP;
 LP:t   = koff2*LSAE - kon2*LE*LP - Gp*LP;
 LE     = Etot - LSAE;

 /* MID DOSE MODEL
    The data for the mid dose model was  taken from L Aarons, et al.,
    the last eleven points of the orally administered does.

11/14 /user2/garyr/Documents/SalicylicAcid.odt



*/
 //     MID DOSE PARAMETER
 real MSAtot = 73.378 mg/L,              // Total MID Dose concentration
 //     MID DOSE MODEL VARIABLES
      MSA(t)  mg/L,                      // MID dose SA
      MSAE(t)  mg/L,                     // MID dose SA enzyme complex
      ME(t)  mg/L,                       // MID dose free enzyme
      MP(t) mg/L;                        // MID dose product
 //     MID DOSE INITIAL CONDITIONS
 when(t=t.min) {MSA=MSAtot; MSAE=0; MP=0;}
 //     MID DOSE ORDINARY DIFFERENTIAL AND MASS BALANCE EQUATIONS
 MSA:t = -kon1*MSA*ME+koff1*MSAE;
 MSAE:t = kon1*MSA*ME-koff1*MSAE - koff2*MSAE + kon2*ME*MP;
 MP:t   = koff2*MSAE - kon2*ME*MP - Gp*MP;
 ME     = Etot - MSAE;

 /* HIGH DOSE MODEL
    The data from the high does model was taken from LF Prescott, 
et al.,the last nine points from the Control.
*/
 //     HIGH DOSE PARAMETER
 real HSAtot = 343.42 mg/L,              // Total HIGH Dose 
concentration
 //     HIGH DOSE MODEL VARIABLES
      HSA(t)  mg/L,                      // HIGH dose SA
      HSAE(t)  mg/L,                     // HIGH dose SA enzyme complex
      HE(t)  mg/L,                       // HIGH dose free enzyme
      HP(t) mg/L;                        // HIGH dose product
 //     HIGH DOSE INITIAL CONDITIONS
 when(t=t.min) {HSA=HSAtot; HSAE=0; HP=0;}
 //     HIGH DOSE ORDINARY DIFFERENTIAL AND MASS BALANCE EQUATIONS
 HSA:t = -kon1*HSA*HE+koff1*HSAE;
 HSAE:t = kon1*HSA*HE-koff1*HSAE-koff2*HSAE+kon2*HE*HP;
 HP:t   = koff2*HSAE - kon2*HE*HP - Gp*HP;
 HE     =Etot - HSAE;

}
/*
 DETAILED DESCRIPTION:

 This is an enzyme conversion model for salicylic acid clearance
 simultaneously fitting three independent data sets using shared 
parameters.
 The initial concentrations of salicylic acid range over almost two 
orders of
 magnitude.

         kon1->        koff2->     Gp->
  LSA+LE<-----> LSAE <-----> LE+LP-----> (Low Dose) 
         <-koff1        <-kon2

         kon1->        koff2->     Gp->
  MSA+ME<-----> MSAE <-----> ME+MP-----> (Mid Dose)
         <-koff1        <-kon2

         kon1->        koff2->     Gp->
  HSA+HE<-----> HSAE <-----> HE+HP-----> (Hi Dose)
         <-koff1        <-kon2

    koff1=KD1*kon1   koff2=KD2*kon1   

 where LSA, MSA, and HSA are Salicylic Acid concentrations,
 HE, ME, and HE are the free enzyme concentrations,
 LSAE, MSAE, and HSAE are salicylic acid-enzyme complex concentrations, 
and
 LP, MP, and HP are the product concentrations.

The parameters kon1, koff2, koff1, kon2, Gp, and the total amount of 
enzyme 
are the same for all three models. 

 KEY WORDS:
 aspirin, salicylic acid, enzyme, clearance, covariance, confidence 
limits,
 optimization, Data, Tutorial

 REVISION HISTORY:
  Written by Gary M Raymond Aug 2009
  Revised by JBB 3dec2010 to have clearance for product P, add Eqs for 
P,
                          so that system is open. Mass balance holds if 
Gp=0.

 JSim software was developed with support from NIH grants HL088516,
 and HL073598. Please cite these grants in any publication for which

13/14 /user2/garyr/Documents/SalicylicAcid.odt



 this software is used and send one reprint of published abstracts or
 articles to the address given below. Academic use is unrestricted.
 Software may be copied so long as this copyright notice is included.

 Copyright (C) 1999-2011 University of Washington.

 Contact Information:

 The National Simulation Resource,
 Director J. B. Bassingthwaighte,
 Department of Bioengineering,
 University of Washington, Seattle, WA
 98195-5061

 */

